

Introduction to *Mathematica*

ME305 & AM501 1993 Winter
Noboru Kikuchi

■ Matrix and Vector Algebra

■ define a 3-by-4 matrix A

After defining a 3-by-4 matrix A, we output it in the form of a matrix. In *Mathematica*, 1,2,3,... are recognized as symbols rather than numerical values. We shall recognize this feature in the following operation.

```
A={ {1,2,3,4}, {5,4,7,8}, {-4,-3,-2,-1} };  
MatrixForm[A]
```

```
1      2      3      4  
5      4      7      8  
-4     -3     -2     -1
```

■ find the eigenvalues of a square matrix AA^T

We shall compute the eigenvalues of a square matrix obtained by AA^T . To this end, there are two ways to do it. The first is computation of the eigenvalues numerically by defining the matrix numerically using the operation `N[]`. If we do not operate `N[]` on the square matrix, its components are recognized as symbols, and then it takes considerably large amount of algebra to find the eigenvalues without introducing any of "round off or approximation" errors. It is noted that we can compute the eigenvalues for symbols after large computation, but it is not realistic at all.

```
Eigenvalues[N[A.Transpose[A]]]  
Eigenvalues[A.Transpose[A]]
```

```
{201.805, 0.598203, 11.5971}
```

$$\left\{ \frac{214}{3} + \frac{4799 \cdot 2^{5/3}}{3 \left(465160 + i \sqrt{4671963198}\right)^{1/3}} + \frac{2^{4/3} \left(465160 + i \sqrt{4671963198}\right)^{1/3}}{3}, \right.$$

$$\frac{214}{3} + \frac{i}{2} \sqrt{3} \left(\frac{-4799 \cdot 2^{5/3}}{3 \left(465160 + i \sqrt{4671963198}\right)^{1/3}} + \frac{2^{4/3} \left(465160 + i \sqrt{4671963198}\right)^{1/3}}{3} \right) -$$

$$\left(\frac{4799 \cdot 2^{5/3}}{3 \left(465160 + i \sqrt{4671963198}\right)^{1/3}} + \frac{2^{4/3} \left(465160 + i \sqrt{4671963198}\right)^{1/3}}{3} \right) / 2,$$

$$\frac{214}{3} - \frac{i}{2} \sqrt{3} \left(\frac{-4799 \cdot 2^{5/3}}{3 \left(465160 + i \sqrt{4671963198}\right)^{1/3}} + \frac{2^{4/3} \left(465160 + i \sqrt{4671963198}\right)^{1/3}}{3} \right) -$$

$$\left(\frac{4799 \cdot 2^{5/3}}{3 \left(465160 + i \sqrt{4671963198}\right)^{1/3}} + \frac{2^{4/3} \left(465160 + i \sqrt{4671963198}\right)^{1/3}}{3} \right) / 2 \left. \right\}$$

It is certain that not only the eigenvalues but also the eigenvectors of a square matrix can be obtained numerically in *Mathematica*. Further, if we wish to compute both the eigenvalues and eigenvectors, it can be done by using command `Eigenystem[]` rather than `Eigenvalues[]` and `Eigenvectors[]`.

```
Eigenvectors[N[A.Transpose[A]]]  
Eigenystem[N[A.Transpose[A]]]
```

```
{{0.372034, 0.872243, -0.317464},  
{-0.848767, 0.458123, 0.264041},  
{-0.375745, -0.17122, -0.910768}}
```

```
{{201.805, 0.598203, 11.5971},  
{{0.372034, 0.872243, -0.317464},  
{-0.848767, 0.458123, 0.264041},  
{-0.375745, -0.17122, -0.910768}}}
```

■ A Complex Matrix

Using two 3-by-3 matrices A and B, we shall describe a complex matrix AB. "I" in *Mathematica* means Sqrt[-1] for expression of complex numbers. We can extract the real and imaginary portions from general complex numbers/matrices using Re[] and Im[], respectively. Furthermore, the absolute value and argument of each complex component of a matrix can also be computed in *Mathematica* using Abs[] and Arg[] operations, respectively. "%" is identified with the variable obtained (or defined) in the last operation in *Mathematica*, and g/N means that g will be regarded as numbers rather than symbols.

```

A={{1,2,3},{3,2,1},{-1,1,-1}}
B={{1,0,1},{-1,-1,1},{4,1,2}}
AB=A + I B;
MatrixForm[AB]
MatrixForm[Re[AB]]
MatrixForm[Im[AB]]
Abs[AB]
MatrixForm[%/N]
Arg[AB]
MatrixForm[%/N]

{{1, 2, 3}, {3, 2, 1}, {-1, 1, -1}}
{{1, 0, 1}, {-1, -1, 1}, {4, 1, 2}}

1 + I      2      3 + I
3 - I      2 - I      1 + I
-1 + 4 I   1 + I      -1 + 2 I

1    2    3
3    2    1
-1   1    -1

1    0    1
-1   -1   1
4    1    2

{{Sqrt[2], 2, Sqrt[10]}, {Sqrt[10], Sqrt[5], Sqrt[2]},
 {Sqrt[17], Sqrt[2], Sqrt[5]}}

1.41421    2.      3.16228
3.16228    2.23607  1.41421
4.12311    1.41421  2.23607

{{ $\frac{\text{Pi}}{4}$ , 0, ArcTan[3, 1]}, {ArcTan[3, -1], ArcTan[2, -1],  $\frac{\text{Pi}}{4}$ },
 {ArcTan[-1, 4],  $\frac{\text{Pi}}{4}$ , ArcTan[-1, 2]}}

0.785398    0      0.321751
-0.321751   -0.463648  0.785398
1.81577     0.785398   2.03444

```

■ Vector Algebra

Two three components vectors u and v are defined, then we operate "*", "/", ".", and Outer[Times,u,v]. "*" and "/" are operated to each component, while "." means the inner (scalar) product $u^T v$ of two vectors. Outer[Times, u, v] means that uv^T .

```

u={u1,u2,u3}
v={v1,v2,v3}
u*v
u/v
u.v
Outer[Times,u,v]
{u1, u2, u3}
{v1, v2, v3}
{u1 v1, u2 v2, u3 v3}
{u1/v1, u2/v2, u3/v3}
u1 v1 + u2 v2 + u3 v3
{{u1 v1, u1 v2, u1 v3}, {u2 v1, u2 v2, u2 v3},
 {u3 v1, u3 v2, u3 v3}}
```

■ Matrix Algebra

After defining two 3-by-3 matrices A and B symbolically, we make algebra "+", "*", "/", "^" and "." in *Mathematica*. It follows from the output that these operations are applied on each component except "." that assumes matrix multiplication instead of componentwise multiplication. Transpose[A] means A^T . Det[A] is the determinant of a square matrix A, while Inverse[A] is A^{-1} . In this example, we define the inverse of A by a new matrix AI.

```

A={{a11,a12,a13},{a21,a22,a23},{a31,a32,a33}};
MatrixForm[A]
B={{b11,b12,b13},{b21,b22,b23},{b31,b32,b33}};
MatrixForm[B]
A+B
A*B
A/B
A.B
Transpose[A]
A^2
A^(-1)
Det[A]
AI=Inverse[A]
```

$$\begin{array}{lll} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{array}$$

$$\begin{array}{lll} b_{11} & b_{12} & b_{13} \\ b_{21} & b_{22} & b_{23} \\ b_{31} & b_{32} & b_{33} \end{array}$$

$$\{\{a_{11} + b_{11}, a_{12} + b_{12}, a_{13} + b_{13}\}, \\ \{a_{21} + b_{21}, a_{22} + b_{22}, a_{23} + b_{23}\}, \\ \{a_{31} + b_{31}, a_{32} + b_{32}, a_{33} + b_{33}\}\}$$

$$\{\{a_{11} b_{11}, a_{12} b_{12}, a_{13} b_{13}\}, \{a_{21} b_{21}, a_{22} b_{22}, a_{23} b_{23}\}, \\ \{a_{31} b_{31}, a_{32} b_{32}, a_{33} b_{33}\}\}$$

$$\{\{\frac{a_{11}}{b_{11}}, \frac{a_{12}}{b_{12}}, \frac{a_{13}}{b_{13}}\}, \{\frac{a_{21}}{b_{21}}, \frac{a_{22}}{b_{22}}, \frac{a_{23}}{b_{23}}\}, \{\frac{a_{31}}{b_{31}}, \frac{a_{32}}{b_{32}}, \frac{a_{33}}{b_{33}}\}\}$$

$$\{\{a_{11} b_{11} + a_{12} b_{21} + a_{13} b_{31}, a_{11} b_{12} + a_{12} b_{22} + a_{13} b_{32}, \\ a_{11} b_{13} + a_{12} b_{23} + a_{13} b_{33}\}, \\ \{a_{21} b_{11} + a_{22} b_{21} + a_{23} b_{31}, \\ a_{21} b_{12} + a_{22} b_{22} + a_{23} b_{32}, a_{21} b_{13} + a_{22} b_{23} + a_{23} b_{33}\} \\ , \{a_{31} b_{11} + a_{32} b_{21} + a_{33} b_{31}, \\ a_{31} b_{12} + a_{32} b_{22} + a_{33} b_{32}, a_{31} b_{13} + a_{32} b_{23} + a_{33} b_{33}\} \\ \}$$

$$\{\{a_{11}, a_{21}, a_{31}\}, \{a_{12}, a_{22}, a_{32}\}, \{a_{13}, a_{23}, a_{33}\}\}$$

$$\{\{a_{11}^2, a_{12}^2, a_{13}^2\}, \{a_{21}^2, a_{22}^2, a_{23}^2\}, \{a_{31}^2, a_{32}^2, a_{33}^2\}\}$$

$$\{\{\frac{1}{a_{11}}, \frac{1}{a_{12}}, \frac{1}{a_{13}}\}, \{\frac{1}{a_{21}}, \frac{1}{a_{22}}, \frac{1}{a_{23}}\}, \{\frac{1}{a_{31}}, \frac{1}{a_{32}}, \frac{1}{a_{33}}\}\}$$

$$-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - a_{11} a_{23} a_{32} - \\ a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}$$

$$\begin{aligned}
& \{ \{ (-(a_{23} a_{32}) + a_{22} a_{33}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (a_{13} a_{32} - a_{12} a_{33}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (-(a_{13} a_{22}) + a_{12} a_{23}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}) \}, \\
& \{ (a_{23} a_{31} - a_{21} a_{33}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (-(a_{13} a_{31}) + a_{11} a_{33}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (a_{13} a_{21} - a_{11} a_{23}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}) \}, \\
& \{ (-(a_{22} a_{31}) + a_{21} a_{32}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (a_{12} a_{31} - a_{11} a_{32}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}), \\
& \quad (-(a_{12} a_{21}) + a_{11} a_{22}) / \\
& \quad (-(a_{13} a_{22} a_{31}) + a_{12} a_{23} a_{31} + a_{13} a_{21} a_{32} - \\
& \quad a_{11} a_{23} a_{32} - a_{12} a_{21} a_{33} + a_{11} a_{22} a_{33}) \} \}
\end{aligned}$$

If components of a matrix are rational functions, then we can pick up only their denominator or/and numerator. Indeed, in *Mathematica*

Denominator[AI]

Numerator[AI]

```
{
  {-(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33,
   -(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33,
   -(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33},
  {-(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33,
   -(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33,
   -(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33},
  {-(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33,
   -(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33},
  {-(a13 a22 a31) + a12 a23 a31 + a13 a21 a32 -
   a11 a23 a32 - a12 a21 a33 + a11 a22 a33}}
  {{-(a23 a32) + a22 a33, a13 a32 - a12 a33,
   -(a13 a22) + a12 a23},
  {a23 a31 - a21 a33, -(a13 a31) + a11 a33,
   a13 a21 - a11 a23}, {-(a22 a31) + a21 a32,
   a12 a31 - a11 a32, -(a12 a21) + a11 a22}}
```

■ Solution of Algebraic Equations

We can solve a system of linear (or even nonlinear) equations symbolically in *Mathematica* using Solve command. If a single equation is considered, we can express it as a single equation. If several equations must be considered, then these should be expressed in a form of list (i.e. vector) in *Mathematica*. After setting up equations, we must specify the variables we would like to find. In the following examples, we have two equations, and they are solved for x1 and x2.

```
Solve[{a11*x1+a12*x2==b1,
       a21*x1+a22*x2==b2},
       {x1,x2}]
  {{x1 -> - (
    -((-a12 a21) + a11 a22) b1) - a12 (a21 b1 - a11 b2)
    a11 (-a12 a21) + a11 a22) \
  , x2 -> - (a21 b1 - a11 b2)
    -a12 a21) + a11 a22}}
```

■ Functions and Their Graphical Representation

■ Definition of a Function

We shall look at capability of *Mathematica* for function operations and graphic capability. First two functions f and g are defined so that we can substitute any expression to the argument.

```
f[x_]:=x-2 x+x^2 + x Cos[Exp[-x]]
g[y_]:=y+y^2 Sin[1/(1+y^2)]
```

■ Operation of a Function

Using these two functions, we shall define a new function $f(xy)+g(x+y)$ that is a function of x and y . Then we shall express this function as a polynomial of x whose coefficients are functions of y by using `Collect[h,x]` command in *Mathematica*. The same function is also collected in y instead of x .


```
Collect[f[x y] + g[x+y],x]
Collect[f[x y] + g[x+y],y]
Coefficient[f[x y] + g[x+y],x^2]
```

$$\begin{aligned}
 & y + y^2 \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right] + x^2 \left(y^2 + \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right]\right) + \\
 & x (1 - y + y \operatorname{Cos}[E^{-(x + y)}]) + 2 y \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right] \\
 & x + x^2 \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right] + y^2 \left(x^2 + \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right]\right) + \\
 & y (1 - x + x \operatorname{Cos}[E^{-(x + y)}]) + 2 x \operatorname{Sin}\left[\frac{1}{1 + (x + y)^2}\right] \\
 & y^2
 \end{aligned}$$


■ Graphics

The functions f and g are defined in arbitrary arguments x and y by adding $_$ after x and y , we can define functions in other variable, say z . We shall plot the functions $f(z)$ and $g(z)$ in the interval $(-4,4)$ using Plot command. The third command ParametricPlot draws a curve defined on the two dimensional plane with a parameter x in $(-4,4)$. A new function fg is then defined by $f(p)g(1+p-q)$ that is a function of p and q . The profile of this function is plotted on the domain $(-2,2) \times (-2,2)$. "PlotRange -> All" in Plot command means that the whole profile of the function is plotted without truncation of the function. "PlotPoints -> 20" in Plot command means that 20 subintervals are assumed in each direction to define the profile. If this number is increased, we can expect smooth profile. For most of functions, it can be utilized the default value for PlotPoints. The last ContourPlot is for drawing contour lines with density that shows the value of the function.


```
Plot[f[z],{z,-4,4}]
Plot[g[z],{z,-4,4}]
ParametricPlot[{f[x],g[x]},{x,-4,4}]
fg=f[p q] g[1+p-q]
Plot3D[fg,{p,-2,2},{q,-2,2},
      PlotRange -> All,
      PlotPoints -> 20]
ContourPlot[fg,{p,-2,2},{q,-2,2},
      PlotPoints -> 30]
```



-Graphics-




-Graphics-




-Graphics-

$$\frac{-(p - q) + p^2 q^2 + p q \cos[E^{-(p - q)}]}{(1 + p - q + (1 + p - q)^2 \sin[\frac{1}{1 + (1 + p - q)^2}]})$$



-SurfaceGraphics-



-ContourGraphics-

It is possible to find a root, i.e. a solution of a nonlinear equation defined by a function. Indeed, FindRoot command, see the first line, finds a root of $f(x)=5$ in the vicinity of $x=1$. The second line computes a root in the vicinity of $x=-2$. *Mathematica* also tries to find the minimum value and a minimizer in the vicinity of a specified point, say $(p, q) = (-1, 0)$ using Newton's method. It often fails to find the minimum and its associated minimizer.

■ Root and Minimum of a Function

```
FindRoot[f[x]==5,{x,1}]  
FindRoot[f[x]==5,{x,-2}]  
FindMinimum[fg,{p,-1},{q,0}]  
FindMinimum[f[x],{x,-4}]
```

```
{x -> 2.23891}
```

```
{x -> -1.99411}
```

```
FindMinimum::fmgz:
```

```
Warning: FindMinimum encountered a vanishing gradient.  
The result returned may not be a minimum; it may be a  
maximum or a saddle point.
```

```
{0., {p -> -1., q -> 0.}}
```

```
{16.2795, {x -> -4.03448}}
```

■ Calculus and Flow Commands

After redefining the function f , we output this new function in the three forms :

1. input form in MATHEMATICA
2. FORTRAN form
3. TeX form

Depending on a specific purpose, we can output variables in these three forms. For example, it might be possible to write a FORTRAN program using *Mathematica* as shown in below. If complex equations must be written in a paper using TeX, *Mathematica* may help to write the equation in TeX format. The fifth statement `f[x]>>output.m` means that the variable (or function) $f(x)$ is output into a file `output.m`. We can integrate a given function on a specified interval (or domain) numerically using `NIntegrate` command. Next we define a list (array) "difference" with all zero entries using `Table` command in *Mathematica* . Three component array is generated here. Now we wish to develop a small program using *Mathematica* commands. To do this, we must declare `Block[.....]` to specify the range of program. To control flow of execution, a local index variable "i" is defined by `{i}`. If there are two local indices to control flow, then we must declare, e.g., `{i,j}`. `Do[.....]` in *Mathematica* is the same to DO loop in FORTRAN. In this example, we make the loop equivalent to

```
DO 100 i=1,3
```

```
.....
```

```
100 CONTINUE
```

in FORTRAN. Taylor's series of a given function can be taken by specifying the evaluation point and the number of terms using `Serires[]` command in *Mathematica*. In the following example, we expand the function $f(x)$ in $2*i$ terms, $i = 1, 2,$ and 3 . `Serires[]` expresses Taylor's expansion with the order of "error", thus in order to use this result as a new function or variable, we must drop the `O[]` term. To do this, we use `Normal[]` command. After computing the integral of the Taylor series by using `NIntegrate` command, we compute the difference of these from the integration of the original function. Then `ListPlot[]` draws the graph of the list (or array) of "difference," while we append this result in the existing output file "output.m" using `<<` `<` command. The last statement defined by `!!file` means show the content of file "output.m."

```

f[x_]:=x-2 x+x^2 + Cos[Exp[-x]];
InputForm[f[x]]
FortranForm[f[x]]
TeXForm[f[x]]
f[x]>>output.m
Iexact=NIntegrate[f[x],{x,-2,2}]
difference=Table[0,{i,1,3}]
Block[{i},
  Do[fi=Series[f[x],{x,0,2*i}];
    Print["Taylor's Series (",2*i,") = ",fi];
    Iseries=Integrate[Normal[fi],{x,-2,2}];
    Print["Iseries = ",Iseries];
    difference[[i]]=Iexact-Iseries;
    Print["difference = ",N[difference[[i]],2*i]],
    {i,1,3}]
]
ListPlot[difference]
N[difference,10]>>>output.m
!!output.m

-x + x^2 + Cos[E^(-x)]
-x + x**2 + Cos(E**(-x))
-x + {x^2} + \cos ({e^{-x}})
6.87039
{0, 0, 0}

```

Taylor's Series (2) = $\text{Cos}[1] + (-1 + \text{Sin}[1]) x +$

$$\left(1 - \frac{\text{Cos}[1]}{2} - \frac{\text{Sin}[1]}{2}\right) x^2 + O[x]^3$$

$$\text{Iseries} = 4 \text{Cos}[1] + \frac{8 (2 - \text{Cos}[1] - \text{Sin}[1])}{3}$$

difference = 3.1

Taylor's Series (4) = $\text{Cos}[1] + (-1 + \text{Sin}[1]) x +$

$$\left(1 - \frac{\text{Cos}[1]}{2} - \frac{\text{Sin}[1]}{2}\right) x^2 + \frac{\text{Cos}[1] x^3}{2} +$$

$$\left(\frac{-\text{Cos}[1]}{4} + \frac{5 \text{Sin}[1]}{24}\right) x^4 + O[x]^5$$

$$\text{Iseries} = 4 \text{Cos}[1] - \frac{8 (6 \text{Cos}[1] - 5 \text{Sin}[1])}{15} +$$

$$\frac{8 (2 - \text{Cos}[1] - \text{Sin}[1])}{3}$$

difference = 2.546

Taylor's Series (6) = $\text{Cos}[1] + (-1 + \text{Sin}[1]) x +$

$$\left(1 - \frac{\text{Cos}[1]}{2} - \frac{\text{Sin}[1]}{2}\right) x^2 + \frac{\text{Cos}[1] x^3}{2} +$$

$$\left(\frac{-\text{Cos}[1]}{4} + \frac{5 \text{Sin}[1]}{24}\right) x^4 + \left(\frac{\text{Cos}[1]}{24} - \frac{23 \text{Sin}[1]}{120}\right) x^5 +$$

$$\left(\frac{11 \text{Cos}[1]}{240} + \frac{37 \text{Sin}[1]}{360}\right) x^6 + O[x]^7$$


$$\text{Iseries} = 4 \text{Cos}[1] - \frac{8 (6 \text{Cos}[1] - 5 \text{Sin}[1])}{15} +$$

$$\frac{8 (2 - \text{Cos}[1] - \text{Sin}[1])}{3} + \frac{16 (33 \text{Cos}[1] + 74 \text{Sin}[1])}{315}$$

difference = -1.52289

$-x + x^2 + \text{Cos}[E^{(-x)}]$

{3.060580652256568187, 2.545625404880224605, -1.522886542773512646}



-Graphics-

■ Differentiation, Integration, and Fourier Series

We can differentiate and integrate a function in *Mathematica*.

```
D[f[x],x]
```

$$-1 + 2x + \frac{\text{Sin}[E^{-x}]}{E^x}$$

```
Integrate[f[x],{x,0,1}]
```

General::intinit: Loading integration packages.

$$-\left(\frac{1}{6}\right) + \text{CosIntegral}[1] - \text{CosIntegral}\left[\frac{1}{E}\right]$$

```
fiexact=%
```

$$-\left(\frac{1}{6}\right) + \text{CosIntegral}[1] - \text{CosIntegral}\left[\frac{1}{E}\right]$$

For most of functions, analytic integration takes considerably a lot of time and memory to evaluate in *Mathematica*, and may not be so practical. If the exact form of integration is not required, numerical integration could be a good replacement. To this end NIntegrate command can be used. As shown in the following example, NIntegrate implies really small error.

```
fiquadrature=NIntegrate[f[x],{x,0,1}]
N[fiexact]-fiquadrature
```

```
0.627165
```

```
-1.13841 10-18
```

It is certain that we can define some special functions such as the step function using IF[.....] command in *Mathematica*. In the following example, we define a step function, then derive its Fourier series using NIntegrate[...] and Sum[...] commands. Summation is taken in odd


number i whose range is $(1, \dots, n)$ by specifying $\{i, 1, n, 2\}$. The last 2 indicates summation is taken for every other term.

```
stepfunction[x_, a_] := If[x < a, 0, 1]
f = stepfunction[x, 0]
n = 10;
fsn = NIntegrate[f, {x, -Pi, Pi}]/(2 Pi) +
Sum[NIntegrate[f*Sin[i*x], {x, -Pi, Pi}]*Sin[i*x]/Pi,
  {i, 1, n, 2}]
Plot[fsn, {x, -Pi, Pi}]
If[x < 0, 0, 1]

$$\frac{1.5708}{\text{Pi}} + \frac{2. \text{Sin}[x]}{\text{Pi}} + \frac{0.666667 \text{Sin}[3 x]}{\text{Pi}} + \frac{0.4 \text{Sin}[5 x]}{\text{Pi}} +$$


$$\frac{0.285714 \text{Sin}[7 x]}{\text{Pi}} + \frac{0.222222 \text{Sin}[9 x]}{\text{Pi}}$$

```



-Graphics-

■ Curve Fitting

Using *Mathematica* we can make curve fitting of a set of discrete data. For the function

$$f(x) = e^x \sin(2\pi x)$$

we shall define a data set by

$$\left\{ x_i = -1 + \frac{2(i-1)}{n}, f(x_i) \right\}, \quad i = 1, 2, 3, \dots, n+1$$

for a large n , say 100. Then this data set is fitted using the following 5 basis functions

$$\{ \exp(x) \quad \sin(x) \quad x \quad x^2 \quad x^3 \}$$

That is, the data is fitted by a linear combination of these basis functions :


$$f(x) \approx c_1 \exp(x) + c_2 \sin(x) + c_3 x + c_4 x^2 + c_5 x^3$$

by the least squares method : the coefficient c_1, \dots, c_5 are determined so as to minimize the residual

$$R = \frac{1}{2} \sum_{i=1}^{n+1} \{f(x_i) - c_1 \exp(x_i) + c_2 \sin(x_i) + c_3 x_i + c_4 x_i^2 + c_5 x_i^3\}^2$$


In *Mathematica* we use Fit[.....] command to make the best curve fitting as follows :

```
f[i_,n_] := Exp[-1+2*(i-1)/n]*Sin[2*Pi*(-1+2*(i-1)/n)]
n=100;
data=Table[{-1+2*(i-1)/n,f[i,n]},{i,1,n+1}];
g=ListPlot[data]
fi=Fit[data,{Exp[x],Sin[x],x,x^2,x^3},x]
h=Plot[fi,{x,-1,1}]
Show[g,h]
```




-Graphics-

```
0.143574 Ex - 2019.33 x - 1.02653 x2 + 316.868 x3 +
2023.6 Sin[x]
```



-Graphics-



-Graphics-


Since the above choice of the basis functions is not quite effective, we shall make curve fitting using the basis functions of a m -th degree complete polynomial :

$$f(x) \approx \sum_{j=0}^m c_j x^j$$


For $m=10$, we have very accurate curve fitting as follows :

```
fi=Fit[data,Table[x^j,{j,0,10}],x]  
h=Plot[fi,{x,-1,1}]  
Show[g,h]
```

$$0.00173229 + 6.2778 x + 6.14739 x^2 - 38.1008 x^3 - \\ 38.5752 x^4 + 60.7599 x^5 + 66.7642 x^6 - 37.2472 x^7 - \\ 46.2662 x^8 + 8.31181 x^9 + 11.9312 x^{10}$$



-Graphics-




-Graphics-

■ Examples in ME210

■ Bending Moment and Shear Force Diagrams

We shall apply Mathematica to solve problems in ME210. As the first example, let us draw the bending moment and shear force diagrams of the cantilever subject to a given set of loading and support conditions as shown in the following figure. A distributed load of 3 kips/ft extends over 8 ft of the beam and the 10 kip load is applied at the attached short beam.



In this case, the bending moment distribution M and shear force diagram V becomes

$$M(x) = \begin{cases} -(p_d x) \frac{x}{2} & x \leq a \\ -(p_d a) \left(x - a + \frac{a}{2} \right) & a \leq x \leq b \\ -(p_d a) \left(x - a + \frac{a}{2} \right) + P_p c - P_p (x - a - b) & b \leq x \end{cases}$$

and

$$V(x) = \begin{cases} -p_d x & x \leq a \\ -p_d a & a \leq x \leq b \\ -p_d a - P & b \leq x \end{cases}$$

respectively. Noting that

$$-(p_d a) \left(x - a + \frac{a}{2} \right) = -(p_d x) \frac{x}{2} + \frac{1}{2} \{ (p_d x)x - 2(p_d a)x + (p_d a)a \} = -(p_d x) \frac{x}{2} + \frac{1}{2} p_d (x - a)^2$$


we can set up the following Mathematica program to draw the bending moment and shear force diagrams :

```


a=8;
b=3;
c=2;
d=3;
Pd=3;
Pp=10;
M=- (Pd*x)*(x/2)+
  ((Pd*(x-a))*(x-a)/2)*If[x<a,0,1]+
  Pp*c*If[x<a+b,0,1]-
  Pp*(x-a-b)*If[x<a+b,0,1]
V=-Pd*x+
  Pd*(x-a)*If[x<a,0,1]-
  Pp*If[x<a+b,0,1]
Plot[M,{x,0,a+b+c+d}]
Plot[V,{x,0,a+b+c+d}]
L=a+b+c+d
M/.{x->L}
V/.{x->L}

```

$$\begin{aligned}
 & \frac{-3x^2}{2} + \frac{3(-8+x)^2}{2} \text{If}[x < a, 0, 1] + \\
 & 20 \text{If}[x < a+b, 0, 1] - \\
 & 10(-11+x) \text{If}[x < a+b, 0, 1] \\
 & -3x + 3(-8+x) \text{If}[x < a, 0, 1] - \\
 & 10 \text{If}[x < a+b, 0, 1]
 \end{aligned}$$



-Graphics-



-Graphics-

16


-318

-34

■ Reaction Force and Deflection of a Beam Subject to a Distributed Force

We shall compute reaction and deflection of the statically indeterminate beam subject to a linearly distributed load as shown in the following figure. The beam is fixed at the right end, while it is supported by a hinge at the left end. Let the length of the beam is L , and the x coordinate is taken from the left end point along the beam axis. Young's modulus and the moment of inertia of the beam cross section are assumed to be constant over the beam. Here w_0 represents the magnitude of the distributed load at the right end point.

E = Young's modulus
 I = Moment of inertia



Noting that the bending moment distribution is given by

$$M(x) = R_a x - \left(\frac{1}{2} \left(\frac{w_0}{L} x \right) x \right) \frac{x}{3}$$

where R_a is the unknown reaction at the hinge support, the beam deflection y is defined by the following beam equation

$$EI \frac{d^2 y}{dx^2} = M(x) = R_a x - \left(\frac{1}{2} \left(\frac{w_0}{L} x \right) x \right) \frac{x}{3}$$

Integrating this twice and applying the support condition

$$y(0) = y(L) = \frac{dy}{dx}(L) = 0$$

we can obtain the deflection as follows :

$$\frac{dy}{dx} = \frac{1}{EI} \left(\int M(x) dx + c_1 \right)$$

$$y = \int \frac{dy}{dx} dx + \frac{1}{EI} c_2$$

Then apply the three support conditions, Solve[...] in *Mathematica* yields c_1 , c_2 and R_a .

```
Clear[L]
M=(Ra*x-((1/2)*w0*(x/L)*x)*(x/3))/EI
dydx=Simplify[Integrate[M,x]+C1/EI]
y=Simplify[Integrate[dydx,x]+C2/EI]
y0=y/.{x->0}
yL=y/.{x->L}
dydxL=dydx/.{x->L}
Solve[{y0==0,yL==0,dydxL==0},{C1,C2,Ra}]
```

$$\frac{Ra x - \frac{w_0 x^3}{6 L}}{EI}$$

$$\frac{24 C_1 L + 12 L Ra x^2 - w_0 x^4}{24 EI L}$$

$$\frac{120 C_2 L + 120 C_1 L x + 20 L Ra x^3 - w_0 x^5}{120 EI L}$$

$$\frac{C_2}{EI}$$

$$\frac{120 C_2 L + 120 C_1 L^2 + 20 L^4 Ra - L^5 w_0}{120 EI L}$$

$$\frac{24 C_1 L + 12 L^3 Ra - L^4 w_0}{24 EI L}$$


$$\left\{ \left\{ C_1 \rightarrow \frac{-(L^3 w_0)}{120}, C_2 \rightarrow 0, Ra \rightarrow \frac{L w_0}{10} \right\} \right\}$$

■ Reaction Force and Deflection of a Beam Subject to a Point Force

We shall obtain reaction and deflection of a statically indeterminate beam that is subject to a point force P at the distance a from the left end point where the beam is supported by a hinge, and is fixed at the right end. If the x coordinate is taken from the left end, the support condition of the beam is given by

$$y(0) = 0 \quad \text{and} \quad y(L) = \frac{dy}{dx}(L) = 0$$

where y is the deflection of the beam.



If the reaction at the hinge support is denoted by R_a , the bending moment distribution becomes

$$M(x) = \begin{cases} R_a x & \text{for } x < a \\ R_a x - P(x - a) & \text{for } a \leq x \end{cases}$$

Thus, we can obtain the deflection of the beam by integrating the beam equation twice :

$$EI \frac{d^2 y}{dx^2} = M(x) = \begin{cases} R_a x & \text{for } x < a \\ R_a x - P(x - a) & \text{for } a \leq x \end{cases}$$

Noting that we have two different description of the bending moment distribution in the beam equation, we must integrate the second order differential equation separately in two different intervals $(0, a)$ and (a, L) , and then we must apply the continuity condition on the deflection as well as the slope of the deflection at the interface $x = a$:

$$\lim_{\varepsilon \rightarrow 0} y(a - \varepsilon) = y(a_-) = y(a_+) = \lim_{\varepsilon \rightarrow 0} y(a + \varepsilon)$$

$$\lim_{\varepsilon \rightarrow 0} \frac{dy}{dx}(a - \varepsilon) = \frac{dy}{dx}(a_-) = \frac{dy}{dx}(a_+) = \lim_{\varepsilon \rightarrow 0} \frac{dy}{dx}(a + \varepsilon)$$

We shall integrate the beam equation separately in the two different interval using *Mathematica*. We first integrate the beam equation twice :

$$\frac{dy_1}{dx} = \frac{1}{EI} \left\{ \int R_a x dx + c_1 \right\}$$

$$y_1 = \int \frac{dy_1}{dx} dx + \frac{1}{EI} c_2$$

$$\frac{dy_2}{dx} = \frac{1}{EI} \left\{ \int (R_a x - P(x-a)) dx + c_3 \right\}$$

$$y_2 = \int \frac{dy_2}{dx} dx + \frac{1}{EI} c_4$$

where c_1 , c_2 , c_3 , and c_4 are constants generated by integration which must be determined by the continuity and support conditions. Using Solve[.....] command for five conditions with respect to five unknowns, we can find c_1 , c_2 , c_3 , c_4 , and R_a as follows :

```

M1=Ra*x/EI
dy1dx=Integrate[M1,x]+C1/EI
y1=Integrate[dy1dx,x]+C2/EI
M2=(Ra*x-P*(x-a))/EI
dy2dx=Integrate[M2,x]+C3/EI
y2=Integrate[dy2dx,x]+C4/EI
y0=y1/.{x->0}
yL=y2/.{x->L}
dydxL=dy2dx/.{x->L}
jdydxa=(dy2dx-dy1dx)/. {x->a}
jya=(y2-y1)/. {x->a}
Solve[{y0==0,yL==0,dydxL==0,jdydxa==0,jya==0},
      {C1,C2,C3,C4,Ra}]
{C1,C2,C3,C4,Ra}=Simplify[{C1,C2,C3,C4,Ra}/. %//First]

```

$$\frac{Ra \ x}{EI}$$

$$\frac{C1}{EI} + \frac{Ra \ x^2}{2 \ EI}$$

$$\frac{C2}{EI} + \frac{C1 \ x}{EI} + \frac{Ra \ x^3}{6 \ EI}$$

$$\frac{Ra \ x - P \ (-a + x)}{EI}$$

$$\frac{C3}{EI} + \frac{a \ P \ x}{EI} + \frac{(-P + Ra) \ x^2}{2 \ EI}$$

$$\frac{C4}{EI} + \frac{C3 \ x}{EI} + \frac{a \ P \ x^2}{2 \ EI} + \frac{(-P + Ra) \ x^3}{6 \ EI}$$

$$\frac{C2}{EI}$$

$$\frac{C4}{EI} + \frac{C3 L}{EI} + \frac{a L^2 P}{2 EI} + \frac{L^3 (-P + Ra)}{6 EI}$$

$$\frac{C3}{EI} + \frac{a L P}{EI} + \frac{L^2 (-P + Ra)}{2 EI}$$

$$-\left(\frac{C1}{EI}\right) + \frac{C3}{EI} + \frac{a^2 P}{EI} - \frac{a^2 Ra}{2 EI} + \frac{a^2 (-P + Ra)}{2 EI}$$

$$-\left(\frac{a C1}{EI}\right) - \frac{C2}{EI} + \frac{a C3}{EI} + \frac{C4}{EI} + \frac{a^3 P}{2 EI} - \frac{a^3 Ra}{6 EI} + \frac{a^3 (-P + Ra)}{6 EI}$$

$$\left\{ \left\{ C1 \rightarrow \frac{a^2 P}{2} + \frac{6 L^3 (2 a L P - L^2 P) + L^2 (-6 a^3 P - 6 (3 a L^2 P - L^3 P))}{24 L^3} \right\} \right.$$

$$, C2 \rightarrow 0, C4 \rightarrow \frac{a^3 P}{6},$$

$$Ra \rightarrow -\left(\frac{2 a L P - L^2 P}{L^2}\right) -$$

$$\frac{6 L^3 (2 a L P - L^2 P) + L^2 (-6 a^3 P - 6 (3 a L^2 P - L^3 P))}{12 L^5},$$

$$C3 \rightarrow \frac{6 L^3 (2 a L P - L^2 P) + L^2 (-6 a^3 P - 6 (3 a L^2 P - L^3 P))}{24 L^3} \left. \right\}$$

$$\left\{ \frac{-(a (-a + L)^2 P)}{4 L}, 0, \frac{-(a (a^2 + L^2) P)}{4 L}, \frac{a^3 P}{6}, P + \frac{a^3 P}{2 L^3} - \frac{3 a P}{2 L} \right\}$$

Thus we have

$$C_1 = -\frac{Pa(L-a)^2}{4L}$$

$$C_2 = 0$$

$$C_3 = -\frac{Pa(L^2 + a^2)}{4L}$$

$$C_4 = \frac{Pa^3}{6}$$

$$R_a = \frac{P(2L^3 + a^3 - L^2a)}{2L^3}$$


Now assuming $a=L/2$, we shall plot the profile of the deflection of the beam by setting $P=1$, $L=1$, and $EI=1$.

```

a=L/2;
y1/.{P->1,L->1,EI->1}
g1=Plot[y1/.{P->1,L->1,EI->1},{x,0,a/.{L->1}}]
y2/.{P->1,L->1,EI->1}
g2=Plot[y2/.{P->1,L->1,EI->1},{x,a/.{L->1},L/.{L->1}}]
Show[g1,g2]


```

$$\frac{-x}{32} + \frac{5x^3}{96}$$




-Graphics-

$$\frac{1}{48} - \frac{5x}{32} + \frac{x^2}{4} - \frac{11x^3}{96}$$



-Graphics-




-Graphics-

■ Examples for ME311

■ Application of the Castigliano Theorem

E = Young's modulus

I = Moment of inertia



We shall consider the same example for the statically indeterminate beam bending with the linearly distributed load with the hinge support at the left end ($x=0$) and the fixed support at the right end ($x=L$), see the figure in above. Using this example, the reaction and the amount of the rotation of the beam are obtained at $x=0$. Noting the second Castigliano theorem yields the deflection of the beam at the left end support ($x=0$), the reaction R_a at this hinge support can be obtained by solving the relation of the zero deflection at $x=0$:

$$y_a = y(0) = \frac{\partial U}{\partial R_a} = \frac{\partial}{\partial R_a} \left[\frac{1}{2EI} \int_0^L M(x)^2 dx \right] = \frac{1}{EI} \int_0^L M \frac{\partial M}{\partial R_a} dx = 0$$

$$\text{where } M(x) = R_a x - \left(\frac{1}{2} w_0 \frac{x}{L} x \right) \frac{x}{3}$$

Translating these into *Mathematica*, we can solve the reaction R_a by using Solve command.

```
M=Ra*x - ((1/2)*w0*(x/L)*x)*(x/3)
ya=Integrate[M*D[M,Ra]/EI,{x,0,L}]
Solve[ya==0,Ra]
```

$$Ra \ x - \frac{w_0 \ x^3}{6 \ L}$$

General::intinit: Loading integration packages.

$$\frac{L^3 \ Ra}{3 \ EI} - \frac{L^4 \ w_0}{30 \ EI}$$

$$\{\{Ra \ \rightarrow \ \frac{L \ w_0}{10}\}\}$$

The slope of the deflection (i.e. rotation of the beam) at the left end hinge support $x=0$, can be obtained by applying the second Castigliano theorem after introducing a fictitious applied moment at the point where we wish to know the rotation. Indeed, if we translate this

$$\theta|_{x=0} = \lim_{M_0 \rightarrow 0} \frac{\partial U}{\partial M_0} = \lim_{M_0 \rightarrow 0} \frac{\partial}{\partial M_0} \left[\frac{1}{2EI} \int_0^L M(x)^2 dx \right]$$

$$\text{where } M(x) = -M_0 + R_a x - \left(\frac{1}{2} w_0 \frac{x}{L} x \right) \frac{x}{3}$$

into *Mathematica* commands, we can obtain the slope of the deflection at $x=0$ as follows :

$$M = -M_0 + R_a x - \left(\frac{1}{2}\right) w_0 \left(\frac{x}{L}\right) x \left(\frac{x}{3}\right)$$

$$\theta_{a0} = \text{Integrate}[M * D[M, M_0] / EI, \{x, 0, L\}] /. \{M_0 \rightarrow 0\}$$


$$\theta_{a0} = \theta_{a0} /. \{R_a \rightarrow L * w_0 / 10\}$$

$$-M_0 + R_a x - \frac{w_0 x^3}{6 L}$$

$$-\frac{(L^2 R_a)}{2 EI} + \frac{L^3 w_0}{24 EI}$$

$$-\frac{(L^3 w_0)}{120 EI}$$

■ Application of the Rayleigh-Ritz Method (1)



We shall obtain the deflection of the beam supported by a hinge at the left end and also fixed at the other end subject to a point load P at the center. The deflection of this beam was already obtained by integrating the second order differential equation written in terms of the bending moment M and the second derivative of the deflection as we have studied in ME210. We shall solve this problem by using the energy method that is one of the most important subjects covered in ME311. More precisely, we shall compute an approximated deflection, that should be very close to the exact solution, by the Rayleigh-Ritz method based on the principle of the minimum potential energy: the deflection in equilibrium minimizes the total potential energy, i.e., the total potential energy attains its minimum at the deflection that yields equilibrium of a structure. The total potential energy consists of the total strain energy and work potential:

$$F = U + W = \frac{1}{2} \int_0^L EI \left(\frac{d^2 w}{dx^2} \right)^2 dx - (-P)w \left(\frac{L}{2} \right)$$

Note that the positive direction of the deflection is up-ward vertical direction, and the applied point force P is down-ward vertical. Thus, when we form the work potential, we must consider the applied force with the negative sign. The principle of the minimum potential energy says the deflection in equilibrium minimizes the total potential energy:

$$\min_w F = \min_w \frac{1}{2} \int_0^L EI \left(\frac{d^2 w}{dx^2} \right)^2 dx - (-P)w \left(\frac{L}{2} \right)$$

Since the support condition must be satisfied with all of possible deflections for equilibrium, minimization of the total potential energy should be considered among deflections satisfying the support condition. In this particular example, zero deflection at the left end, zero deflection and slope at the right end point must be satisfied :

$$w(0) = w(L) = \frac{dw}{dx}(L) = 0$$

If we parameterize possible deflections for equilibrium, e.g., by

$$w(x) = x(x-L)^2 (C_1 + C_2 x + C_3 x^2 + \dots)$$

using "arbitrary" parameters C_1, C_2, C_3, \dots , we may determine these parameters so that the total potential energy is minimized. Then we can define the deflection that yields equilibrium of the structure. If possible deflections are parameterized, the total potential energy becomes a function of these parameters. If a finite number of parameters is used for approximation of a deflection, it becomes a function on a finite number of variables (i.e. parameters). Then the necessary condition of the minimum is vanishing the first derivatives of the function with respect to these variables :

$$\frac{\partial F}{\partial C_i}(C_1, C_2, C_3, \dots) = 0 \quad i = 1, 2, 3, \dots$$

This means that we can derive the same number of equations with the number of parameters introduced. Solving the set of equations with respect to the parameters yields the deflection that minimizes the total potential energy, i.e., in equilibrium. Assuming only three parameters $C_1, C_2,$ and $C_3,$ we shall obtain the deflection using *Mathematica*.


```

w2=x*(x-L)^2(C1+C2*x+C3*x^2)
F=(EI/2)*Integrate[D[D[w2,x],x]^2,{x,0,L}]-
(-P)*w2/.{x->L/2}
F=Expand[F];
Solve[{D[F,C1]==0,D[F,C2]==0,D[F,C3]==0},
{C1,C2,C3}]
w2r=w2/.%//First

```


$$x (-L + x)^2 (C1 + C2 x + C3 x^2)$$

$$\begin{aligned}
& (EI \left(\frac{400 C3^2 L^7}{7} + 4 L^3 (-2 C1 + C2 L)^2 + \right. \\
& \quad 80 C3 L^6 (C2 - 2 C3 L) + \\
& \quad 12 L^3 (-2 C1 + C2 L) (C1 - 2 C2 L + C3 L^2) + \\
& \quad \left. \frac{48 L^5 (3 C2^2 + 5 C1 C3 - 22 C2 C3 L + 17 C3^2 L^2)}{5} + \right. \\
& \quad 4 L^4 (9 C1 C2 - 18 C2^2 L - 28 C1 C3 L + 50 C2 C3 L^2 - \\
& \quad \quad \left. 18 C3^2 L^3) + 4 L^3 \right. \\
& \quad \left. (3 C1^2 - 20 C1 C2 L + 16 C2^2 L^2 + 22 C1 C3 L^2 - \right. \\
& \quad \quad \left. 20 C2 C3 L^3 + 3 C3^2 L^4) \right) / 2 + \\
& \quad \frac{L^3 (C1 + \frac{C2 L}{2} + \frac{C3 L^2}{4}) P}{8}
\end{aligned}$$

$$\{\{C1 \rightarrow \frac{-P}{32 EI}, C2 \rightarrow \frac{-5 P}{64 EI L}, C3 \rightarrow 0\}\}$$

$$x (-L + x)^2 \left(\frac{-P}{32 EI} - \frac{5 P x}{64 EI L} \right)$$

Plot[w2r/.{P->1,EI->1,L->1},{x,0,1}]



-Graphics-

■ Application of the Rayleigh-Ritz Method (2)

Deflection of the beam fixed at the both ends shown in the following figure, will be obtained by the Rayleigh-Ritz method based on the principle of minimum potential energy of a linearly elastic structure. A uniformly distributed load is applied on the right half portion of the beam with intensity p_d . For simplicity, let Young's modulus E be constant, and let the moment of inertia of the cross section be also constant. The length of the beam is described by L . In the Rayleigh-Ritz method, approximated deflection of the beam must satisfy the support condition written in the deflection and its slope. Since the both ends are fixed, the deflection and slope must be zero at these points, i.e.,

$$w(0) = w(L) = \frac{dw}{dx}(0) = \frac{dw}{dx}(L) = 0$$

where w is the deflection, approximations w_i of the deflection may be written by

$$w_i(x) = x^2(x-L)^2(C_1 + C_2x + C_3x^2 + \dots + C_{i+1}x^i)$$

Substitution of this into the form of the total potential energy

$$F = \underbrace{\frac{1}{2} \int_0^L EI \left(\frac{d^2 w}{dx^2}(x) \right)^2 dx}_{\text{total strain energy } U} - \underbrace{\int_{L/2}^L p_d(x) w(x) dx}_{\text{work potential } -W}$$

yields a function F of a finite number of parameters C_1, C_2, \dots, C_{i+1} :

$$F(C_1, C_2, \dots, C_{i+1}) = \frac{1}{2} \int_0^L EI \left(\frac{d^2 w_i}{dx^2}(x) \right)^2 dx - \int_{L/2}^L p_d(x) w_i(x) dx$$

The necessary condition for the minimum of the total potential energy, that is approximated by w_i , is vanishing the gradient of F with respect to the parameters C_1, C_2, \dots, C_{i+1} :

$$\frac{\partial F}{\partial C_1}(C_1, C_2, \dots, C_{i+1}) = 0$$

$$\frac{\partial F}{\partial C_2}(C_1, C_2, \dots, C_{i+1}) = 0$$

.....


$$\frac{\partial F}{\partial C_{i+1}}(C_1, C_2, \dots, C_{i+1}) = 0$$

Solving these $i+1$ number of equations in the $i+1$ parameters C_1, C_2, \dots, C_{i+1} , we can determine the deflection w_i . We shall obtain the three approximated deflections for $i = 0, 1$, and 3 using *Mathematica*.

Young's Modulus = E

Moment of Inertia = I

pd



□ Lowest Order Approximation

```
w0=x^2*(x-L)^2*(C1)
F=(EI/2)*Integrate[D[D[w0,x],x]^2,{x,0,L}]-
  Integrate[pd*w0,{x,L/2,L}];
F=Expand[F];
Solve[D[F,C1]==0,C1]
w0r=w0/.%//First
```

$$C1 x^2 (-L + x)^2$$

$$\{\{C1 \rightarrow \frac{pd}{48 EI}\}\}$$

$$\frac{pd x^2 (-L + x)^2}{48 EI}$$

□ Second Lowest Order Approximation

```
w1=x^2*(x-L)^2*(C1+C2*x)
F=(EI/2)*Integrate[D[D[w1,x],x]^2,{x,0,L}]-
  Integrate[pd*w1,{x,L/2,L}];
F=Expand[F];
Solve[{D[F,C1]==0,D[F,C2]==0},{C1,C2}]
w1r=w1/.%//First
```

$$x^2 (-L + x)^2 (C1 + C2 x)$$

$$\left\{ \left\{ C1 \rightarrow \frac{3 \text{pd}}{256 \text{EI}}, C2 \rightarrow \frac{7 \text{pd}}{384 \text{EI} L} \right\} \right\}$$

$$x^2 (-L + x)^2 \left(\frac{3 \text{pd}}{256 \text{EI}} + \frac{7 \text{pd} x}{384 \text{EI} L} \right)$$

□ Third Order Ritz Approximation

```
w3=x^2*(x-L)^2*(C1+C2*x+C3*x^2+C4*x^3)
F=(EI/2)*Integrate[D[D[w3,x],x]^2,{x,0,L}]-
  Integrate[pd*w3,{x,L/2,L}];
F=Expand[F];
Solve[{D[F,C1]==0,D[F,C2]==0,D[F,C3]==0,D[F,C4]==0},
  {C1,C2,C3,C4}]
w3r=w3/.%//First
```

$$x^2 (-L + x)^2 (C1 + C2 x + C3 x^2 + C4 x^3)$$

$$\left\{ \left\{ C1 \rightarrow \frac{83 \text{pd}}{6144 \text{EI}}, C2 \rightarrow \frac{\text{pd}}{256 \text{EI} L}, C3 \rightarrow \frac{33 \text{pd}}{1024 \text{EI} L^2}, \right. \right.$$


$$\left. \left. C4 \rightarrow \frac{-11 \text{pd}}{512 \text{EI} L^3} \right\} \right\}$$

$$x^2 (-L + x)^2 \left(\frac{83 \text{pd}}{6144 \text{EI}} + \frac{\text{pd} x}{256 \text{EI} L} + \frac{33 \text{pd} x^2}{1024 \text{EI} L^2} - \frac{11 \text{pd} x^3}{512 \text{EI} L^3} \right)$$

□ Graphs of the three approximations

Since the 1st order and the 3rd order Ritz approximations are almost identical, we can say that the deflection of a beam can be obtained using rather few terms of a polynomial with sufficient accuracy.


```
Plot[{w0r/.{pd->1,EI->1,L->1},
      w1r/.{pd->1,EI->1,L->1},
      w3r/.{pd->1,EI->1,L->1}},{x,0,1}]
```



-Graphics-

■ Example of a Plane Frame Structure

In above, we present examples of a beam that can be analyzed by the method studied in ME210. Here we shall provide an example of the plane frame structure, shown in the following figure, that might not be solved by the method studied in ME210.



The frame structure consists of three beams, the length of which is the same, say L . At point A, it is fixed, i.e., the deflection and rotation (the slope of the deflection) are constrained to be zero, while it is supported by a hinge at D. A uniformly distributed load, p_0 , is applied along the

member AB. Since this is a plane frame, we must consider at least bending and axial deformation. Shear effect can be neglected, and torsion is not involved. We shall obtain the reaction forces R_x and R_y at the hinge support D using the energy method, more precisely, by applying the second Castigliano theorem. Noting that point D is supported by a hinge, both x and y components of the displacement are zero, the Castigliano theorem implies that

$$\delta_x = \frac{\partial U^*}{\partial R_x} = 0 \quad \text{and} \quad \delta_y = \frac{\partial U^*}{\partial R_y} = 0$$

where U^* is the total complementary strain energy that must be the same with the total strain energy for any linearly elastic structure :

$$U^* = \sum_{i=1}^n \left\{ \frac{1}{2} \int_0^{L_i} \frac{M_i^2}{EI_i} ds + \int_0^{L_i} \frac{N_i^2}{EA_i} ds \right\}$$

Here n is the total number of members consisting of a given frame structure, L_i is the length, M_i is the bending moment, N_i is the axial force, EI_i is the bending rigidity, EA_i is the rigidity of member i in the axial deformation, and s is the coordinate defined in each member along the beam axis. Noting that the axial force and bending moment are given by

$$N = \begin{cases} -R_y & \text{in member CD} \\ R_x & \text{in member BC} \\ R_y & \text{in member AB} \end{cases}$$

and

$$M = \begin{cases} -R_x s & \text{in member CD} \\ -R_x L - R_y s & \text{in member BC} \\ -R_x L - R_y L + R_x s + (pL_0 s) \frac{s}{2} & \text{in member AB} \end{cases}$$

we can set up the following *Mathematica* program to find the reactions at D :

```

N1=-Ry;
M1=-Rx*s;
N2=Rx;
M2=-Rx*L-Ry*s;
N3=Ry;
M3=-Rx*L-Ry*L+Rx*s+(p0*s)*s/2;
UN=(Integrate[N1^2,{s,0,L}]+
     Integrate[N2^2,{s,0,L}]+
     Integrate[N3^2,{s,0,L}])/(2*EA);
UM=(Integrate[M1^2,{s,0,L}]+
     Integrate[M2^2,{s,0,L}]+
     Integrate[M3^2,{s,0,L}])/(2*EI);
deltax=Simplify[D[UN+UM,Rx]]
deltay=Simplify[D[UN+UM,Ry]]
Solve[{deltax==0,deltay==0},{Rx,Ry}]
{Rx,Ry}=Simplify[{Rx,Ry}/.%//First]

```

$$\frac{L Rx}{EA} + \frac{L^3 (-(L p0) + 40 Rx + 24 Ry)}{24 EI}$$

General::spell1:

Possible spelling error: new symbol name "deltay"
is similar to existing symbol "deltax".

$$\frac{2 L Ry}{EA} + \frac{L^3 (-(L p0) + 6 Rx + 8 Ry)}{6 EI}$$

$$\left\{ \left\{ Rx \rightarrow \frac{L p0}{6} + (2 (3 EI + 2 EA L^2) \right. \right.$$


$$\left. \left. \frac{(-6 EA^2 L^5 p0 + 8 EA L^3 (3 EI + 5 EA L^2) p0)}{(3 EA L^2 (-288 EI^2 - 672 EA EI L^2 - 176 EA^2 L^4))}, \right. \right.$$

$$\left. \left. Ry \rightarrow -\frac{(-6 EA^2 L^5 p0 + 8 EA L^3 (3 EI + 5 EA L^2) p0)}{(-288 EI^2 - 672 EA EI L^2 - 176 EA^2 L^4)} \right\} \right\}$$

$$\left\{ \frac{EA L^3 (3 EI - 4 EA L^2) p0}{4 (18 EI^2 + 42 EA EI L^2 + 11 EA^2 L^4)}, \right.$$

$$\left. \frac{EA L^3 (12 EI + 17 EA L^2) p0}{8 (18 EI^2 + 42 EA EI L^2 + 11 EA^2 L^4)} \right\}$$

■ Examples in ME305



We shall make an analysis of finite element interpolation error of the 4 node quadrilateral element for plane problems using Taylor's expansion of an arbitrary function g in *Mathematica* :

$$g(s_0, t_0) = \underbrace{\sum_{i=0}^2 \sum_{j=0}^2 (s_0 - s)^i (t_0 - t)^j \frac{1}{i!j!} \frac{\partial^{i+j} g}{\partial s^i \partial t^j}(s, t)}_{=g_2(s_0, t_0)} + O\left((s_0 - s)^3, (t_0 - t)^3\right)$$

The shape functions of the 4 node quadrilateral element are

$$N_1(s, t) = \frac{1}{4}(1-s)(1-t) \quad , \quad N_2(s, t) = \frac{1}{4}(1+s)(1-t)$$

$$N_3(s, t) = \frac{1}{4}(1+s)(1+t) \quad , \quad N_4(s, t) = \frac{1}{4}(1-s)(1+t)$$

Then the first derivatives of the interpolation error may be represented by

$$\frac{\partial e_I}{\partial s} = \sum_{i=1}^4 g_2(s_i, t_i) \frac{\partial N_i}{\partial s}(s, t) - \frac{\partial g}{\partial s} \quad , \quad \frac{\partial e_I}{\partial t} = \sum_{i=1}^4 g_2(s_i, t_i) \frac{\partial N_i}{\partial t}(s, t) - \frac{\partial g}{\partial t}$$


```

gT2=Normal[Series[g[s0,t0],{s0,s,2},{t0,t,2}]]
g1=gT2/.{s0->-1,t0->-1};
g2=gT2/.{s0-> 1,t0->-1};
g3=gT2/.{s0-> 1,t0-> 1};
g4=gT2/.{s0->-1,t0-> 1};
gv={g1,g2,g3,g4};
SF={ (1-s)*(1-t)/4,
      (1+s)*(1-t)/4,
      (1+s)*(1+t)/4,
      (1-s)*(1+t)/4}
deds=Simplify[gv.D[SF,s]-Derivative[1,0][g][s,t]]
dedt=Simplify[gv.D[SF,t]-Derivative[0,1][g][s,t]]

```


$$\begin{aligned}
& g[s, t] + (-s + s_0) g^{(1,0)}[s, t] + \\
& \frac{(-s + s_0)^2 g^{(2,0)}[s, t]}{2} + \\
& (-t + t_0) (g^{(0,1)}[s, t] + (-s + s_0) g^{(1,1)}[s, t] + \\
& \frac{(-s + s_0)^2 g^{(2,1)}[s, t]}{2}) + \\
& (-t + t_0)^2 \left(\frac{g^{(0,2)}[s, t]}{2} + \frac{(-s + s_0) g^{(1,2)}[s, t]}{2} + \right. \\
& \left. \frac{(-s + s_0)^2 g^{(2,2)}[s, t]}{4} \right) \\
& (g^{(1,2)}[s, t] - t^2 g^{(1,2)}[s, t] - 2 s g^{(2,0)}[s, t] - \\
& s g^{(2,2)}[s, t] + s t^2 g^{(2,2)}[s, t]) / 2
\end{aligned}$$

General::spell1:

Possible spelling error: new symbol name "dedt" is similar to existing symbol "deds".

$$\begin{aligned}
& (-2 t g^{(0,2)}[s, t] + g^{(2,1)}[s, t] - s^2 g^{(2,1)}[s, t] - \\
& t g^{(2,2)}[s, t] + s^2 t g^{(2,2)}[s, t]) / 2
\end{aligned}$$

For the geometry described in the following figure, we shall consider the case of pure bending characterized by the deformed configuration shown in the figure using the bold face. We shall compute the normal strains, e_x and e_y , and shear strain, g_{xy} , of this deformation, and the normalized strain energies, I_x, I_y , and I_{xy} , due to these three strains with respect to $E/(1-\nu^2)$, where E is Young's modulus and ν is Poisson's ratio. After obtaining these strain energies, we shall plot the ratio I_{xy}/I_x with respect to the element geometry aspect ratio h_y/h_x .



```

xv={0,hx,hx,0}
yv={0,0,hy,hy}
jac={{xv.D[SF,s],yv.D[SF,s]},
      {xv.D[SF,t],yv.D[SF,t]}};
MatrixForm[jac]
jacobian=jac[[1,1]]*jac[[2,2]]-jac[[2,1]]*jac[[1,2]];
jacinv=Simplify[Inverse[jac]];
MatrixForm[jacinv]
uv={-a,a,-a,a}
duds=Simplify[uv.D[SF,s]];
dudt=Simplify[uv.D[SF,t]];
vv={0,0,0,0}
dvds=Simplify[vv.D[SF,s]];
dvdt=Simplify[vv.D[SF,t]];
ex=jacinv[[1,1]]*duds+jacinv[[1,2]]*dudt;
ey=jacinv[[2,1]]*dvds+jacinv[[2,2]]*dvdt;
gxy=jacinv[[1,1]]*dvds+jacinv[[1,2]]*dvdt+
      jacinv[[2,1]]*duds+jacinv[[2,2]]*dudt;
ex=Simplify[ex]
ey=Simplify[ey]
gxy=Simplify[gxy]
Ix=Integrate[(ex^2)*jacobian,{s,-1,1},{t,-1,1}];
Iy=Integrate[(ey^2)*jacobian,{s,-1,1},{t,-1,1}];
Ixy=Integrate[(gxy^2)*jacobian,{s,-1,1},{t,-1,1}];
Ix=Simplify[Ix]
Iy=Simplify[Iy]
Ixy=Simplify[Ixy]
pr=0.3;
d33=(1-pr)/2;
Plot[(d33*Ixy/.{hy->x*hx,a->1})/(Ix/.{hy->x*hx,a->1}),
      {x,0.5,4}]

{0, hx, hx, 0}
{0, 0, hy, hy}

```

$$\frac{hx (1 - t)}{4} + \frac{hx (1 + t)}{4}$$

0

$$\frac{2}{hx} \quad 0$$

$$0 \quad \frac{2}{hy}$$

$$\{-a, a, -a, a\}$$

$$\{0, 0, 0, 0\}$$

$$\frac{-2 a t}{hx}$$


0

$$\frac{-2 a s}{hy}$$

$$\frac{4 a^2 hy}{3 hx}$$

0

$$\frac{4 a^2 hx}{3 hy}$$



-Graphics-